The Mathematica® Journal

In and Out

Edited by Paul Abbott

In and Out offers readers an opportunity to ask questions of the experts. *The Mathematica Journal* encourages readers to submit problems in care of the editor. Answers posted to the *Mathematica* newsgroup, comp.soft-sys.math.mathematica, that appear here are edited for clarity and length.

■ Working with Intervals

Q: A list of times of the form $\{\{t_1, t_1+\delta t_1\}, \{t_2, t_2+\delta t_2\}, \ldots, \{t_n, t_n+\delta t_n\}\}$ represents the periods $\{t_i, t_i+\delta t_i\}$ when a particular system is not working. For two such lists, how can I determine the periods when *both* systems are not working?

A: Carl Woll (carlw@wolfram.com) answers: The function **timelist** simulates nonoverlapping timing data of the form $\{\{t_1, t_1+\delta t_1\}, \{t_2, t_2+\delta t_2\}, \ldots, \{t_n, t_n+\delta t_n\}\}$.

```
timelist[n_] := Function[{a, b}, {a, a + (b − a) Random[]}] @@@
    Partition[Sort[Array[Random[] &, n]], 2, 1]
```

A: Here are two lists of timing data.

```
list1 = timelist[7]
```

$$\begin{pmatrix} 0.000190505 & 0.0190952 \\ 0.0875722 & 0.177131 \\ 0.207103 & 0.210157 \\ 0.214347 & 0.458043 \\ 0.48994 & 0.526966 \\ 0.539981 & 0.84577 \end{pmatrix}$$

list2 = timelist[9]

$$\begin{pmatrix} 0.0376497 & 0.0772187 \\ 0.0773881 & 0.17435 \\ 0.17932 & 0.216258 \\ 0.241548 & 0.34242 \\ 0.399323 & 0.56923 \\ 0.586412 & 0.716589 \\ 0.723289 & 0.759739 \\ 0.776276 & 0.781007 \end{pmatrix}$$

A: After converting each period $\{t_i, t_i + \delta t_i\}$ into an **Interval** using **Apply**, **IntervalIntersection** returns those periods when both systems are not working.

```
IntervalIntersection[Interval @@ list1, Interval @@ list2]
```

```
Interval[{0.0875722, 0.17435}, {0.207103, 0.210157}, {0.214347, 0.216258},
  {0.241548, 0.34242}, {0.399323, 0.458043}, {0.48994, 0.526966}, {0.539981, 0.56923},
  {0.586412, 0.716589}, {0.723289, 0.759739}, {0.776276, 0.781007}]
```

■ HamiltonianCycle

Q: I am interested in counting the number of paths from a node in a network back to that node, counting loops (cycles) only once. Is there any readily available code that does this?

A: Steve Luttrell (steve_usenet@luttrell.org.uk) answers: Use **FindHamiltonianCycle**. Here is an example of how this solves your problem.

A: Here is a random, labeled graph on 10 vertices with one more than half the number of possible edges, which is ususally sufficient to have a Hamiltonian cycle.

```
g = RandomGraph[{10, Ceiling[1 + Binomial[10, 2] / 2]}]
```



A: Using **HighlightGraph**, display a **HamiltonianCycle** of the graph **g**.

```
HighlightGraph[g, FindHamiltonianCycle[g]]
```



Listable Subvalues

Q: I am writing functions for translating graphics primitives. The following function translates a single line object.

```
f[vec_][Line[a_]] := Line[(vec + # &) /@ a]
```

```
b = Line[{{1, 2}, {3, 4}, {5, 6}}];
```

```
f[{1, 3}][b]
```

$$\text{Line}\left[\begin{pmatrix} 2 & 5 \\ 4 & 7 \\ 6 & 9 \end{pmatrix}\right]$$

Q: However, if I have a list of lines it does not work.

```
c = {b, Line[{{2, 3}, {5, 6}, {9, 3}}]};
```

```
f[{1, 3}][c]
```

$$f(\{1, 3\})\left(\left\{\text{Line}\left[\begin{pmatrix} 1 & 2 \\ 3 & 4 \\ 5 & 6 \end{pmatrix}\right], \text{Line}\left[\begin{pmatrix} 2 & 3 \\ 5 & 6 \\ 9 & 3 \end{pmatrix}\right]\right\}\right)$$

Q: What is the best solution?

A: You can map your function over the list of lines.

```
f[{1, 3}] /@ c
```

$$\left\{\text{Line}\left[\begin{pmatrix} 2 & 5 \\ 4 & 7 \\ 6 & 9 \end{pmatrix}\right], \text{Line}\left[\begin{pmatrix} 3 & 6 \\ 6 & 9 \\ 10 & 6 \end{pmatrix}\right]\right\}$$

A: Adding a rule to *f* for the case of list arguments makes this mapping operation automatic.

```
f[vec_][a_List] := f[vec] /@ a
```

```
f[{1, 3}][c]
```

$$\left\{\text{Line}\left[\begin{pmatrix} 2 & 5 \\ 4 & 7 \\ 6 & 9 \end{pmatrix}\right], \text{Line}\left[\begin{pmatrix} 3 & 6 \\ 6 & 9 \\ 10 & 6 \end{pmatrix}\right]\right\}$$

A: David Park (djmp@earthlink.net) provides an alternative solution: define **f[vec_]** as **Function**. The third argument to **Function** is a list of attributes for the purpose of evaluation.

```
Clear[f];

f[vec_] := Function[{a}, Line[(vec + # &) /@ First[a]], {Listable}]

f[{1, 3}][b]
```

$$\text{Line}\left[\begin{pmatrix} 2 & 5 \\ 4 & 7 \\ 6 & 9 \end{pmatrix}\right]$$

```
f[{1, 3}][c]
```

$$\left\{\text{Line}\left[\begin{pmatrix} 2 & 5 \\ 4 & 7 \\ 6 & 9 \end{pmatrix}\right], \text{Line}\left[\begin{pmatrix} 3 & 6 \\ 6 & 9 \\ 10 & 6 \end{pmatrix}\right]\right\}$$

A: Functions for transforming objects in 2 and 3 dimensions are defined in [1] with the packages available at [2].

$f(x) = f(2x) + f(2x+1)$

Q: How can I solve the functional equation $f(x) = f(2x) + f(2x+1)$?

A: The solution to this functional equation is given at mathworld.wolfram.com/-FunctionalEquation.html. Noting that

$$\text{Simplify}\left[1 + \frac{1}{x} == \left(1 + \frac{1}{2x}\right)\left(1 + \frac{1}{2x+1}\right)\right]$$

True

A: then taking logs of both sides, one sees that $f(x) = c \log(1 + 1/x)$, where c is arbitrary, satisfies the functional equation. More generally, since

$$1 + \frac{1}{x} == \prod_{i=1}^{n}\left(1 + \frac{1}{n x + i - 1}\right) \text{ // FullSimplify}$$

True

A: we observe that $f(x) = c \log(1 + 1/x)$ is a solution to the functional equation

$$f(x) = \sum_{i=1}^{n} f(n\,x + i - 1)$$

A: for $n = 2, 3, \ldots$.

A: One way to solve the functional equation is to assume that the asymptotic behavior of the solution is $f(x) \sim c_1/x + c_2\big/x^2 + \cdots$.

$$\mathbf{f_{m\_}(x\_) := \sum_{n=1}^{m} \frac{c_n}{x^n}}$$

A: Substitute this sum into the functional equation and expand into an asymptotic series.

Series[$f(x) - (f(2\,x) + f(2\,x + 1))$ /. $f \to f_6$, {x, ∞, 6}]

$$\frac{c_1 + 2\,c_2}{4\,x^2} + \frac{-c_1 + 2\,c_2 + 6\,c_3}{8\,x^3} + \frac{c_1 - 3\,c_2 + 3\,c_3 + 14\,c_4}{16\,x^4} + \frac{-c_1 + 4\,c_2 - 6\,c_3 + 4\,c_4 + 30\,c_5}{32\,x^5} + \frac{c_1 - 5\,c_2 + 10\,c_3 - 10\,c_4 + 5\,c_5 + 62\,c_6}{64\,x^6} + O\left(\left(\frac{1}{x}\right)^7\right)$$

A: Solving for the coefficients α_n, one sees that $c_n = (-1)^{n+1}\,c_1/n$.

Solve[% == 0, Table[c_n, {n, 2, 6}]]

$$\left\{\left\{c_2 \to -\frac{c_1}{2},\, c_3 \to \frac{c_1}{3},\, c_4 \to -\frac{c_1}{4},\, c_5 \to \frac{c_1}{5},\, c_6 \to -\frac{c_1}{6}\right\}\right\}$$

A: Summing the asymptotic series, one obtains the same solution as earlier, $f(x) = c_1 \log(1 + 1/x)$.

$$\mathbf{\sum_{n=1}^{\infty} \frac{(-1)^{n+1}\,c_1}{n\,x^n}}$$

$$c_1 \log\left(\frac{x + 1}{x}\right)$$

A: Here we apply the same method to $g(x) = g(2\,x + a) + g(2\,x + b)$.

```
Solve[Series[g(x) − (g(2 x + a) + g(2 x + b)) /. g → f4, {x, ∞, 4}] == 0,
    Table[cn, {n, 2, 4}]] // Simplify
```

$$\left\{\left\{c_2 \to -\frac{1}{2}\, c_1\,(a+b),\, c_3 \to \frac{1}{3}\, c_1 \left(a^2 + a\,b + b^2\right),\, c_4 \to -\frac{1}{4}\, c_1 \left(a^3 + a^2\,b + a\,b^2 + b^3\right)\right\}\right\}$$

A: The pattern of the coefficients is clear: the coefficient c_{n+1} is $(-1)^n\, c_1 \sum_{i=0}^{n} a^i\, b^{n-i} \,/\, (n+1)$. Summing the asymptotic series, one obtains the general solution.

$$\text{Simplify}\left[\sum_{n=0}^{\infty} \frac{(-1)^n \sum_{i=0}^{n} a^i\, b^{n-i}\, c_1}{(n+1)\, x^{n+1}}\right]$$

$$\frac{c_1 \left(\log\left(\frac{a+x}{x}\right) - \log\left(\frac{b+x}{x}\right)\right)}{a-b}$$

A: Absorbing the factor $(a-b)$ into the arbitrary constant, the solution can be written as $g(x) = c \log\left(\frac{a+x}{b+x}\right)$. This solution results by taking the logarithm of the following identity.

$$\left(\frac{a+x}{b+x}\right) == \left(\frac{a+2\,x+a}{b+2\,x+a}\right)\left(\frac{a+2\,x+b}{b+2\,x+b}\right) \text{ // Simplify}$$

True

A: Finally, we apply the same method to $h(x) = h(x+1) + h\left(x^2 + x + 1\right)$.

```
Solve[Series[h(x) − (h(x + 1) + h(x^2 + x + 1)) /. h → f10, {x, ∞, 11}] == 0,
   Table[cn, {n, 2, 10}]]
```

$$\left\{\left\{c_2 \to 0,\, c_3 \to -\frac{c_1}{3},\, c_4 \to 0,\, c_5 \to \frac{c_1}{5},\, c_6 \to 0,\, c_7 \to -\frac{c_1}{7},\, c_8 \to 0,\, c_9 \to \frac{c_1}{9},\, c_{10} \to 0\right\}\right\}$$

A: The pattern of the coefficients is clear: the even coefficients vanish, $c_{2\,n} = 0$, and the odd coefficients read $c_{2\,n+1} = (-1)^n\, c_1 \,/\, (2\,n+1)$. Summing the asymptotic series, one obtains the general solution.

$$\text{Sum}\left[\frac{(-1)^{\frac{n-1}{2}}\, c_1}{n\, x^n},\, \{n, 1, \infty, 2\}\right]$$

$$c_1 \tan^{-1}\left(\frac{1}{x}\right)$$

A: See also [3], [4], and [5].

EventLocator

Q: The solution to the differential equation $y'(x) = \cos(x)$ with $y(0) = 0$ is $\sin(x)$.

```
DSolve[{y'(x) == cos(x), y(0) == 0}, y(x), x]
```

$\{\{y(x) \to \sin(x)\}\}$

Q: Solving the same equation numerically, I tried using the event function $y(x) - 1$ to find the extrema of $y(x)$.

```
NDSolve[{y'(x) == cos(x), y(0) == 0}, y, {x, 0, 20},
    Method → {"EventLocator", "Event" → y(x) − 1, "EventAction" :→ Sow[x]}] //
  Reap
```

{{{y → InterpolatingFunction[]}}, {}}

Q: How can I find those points where $y(x) = 1$?

A: Mark Sofroniou (marks@wolfram.com) answers: An event is located when a change in sign in the event function is detected. For the function $\sin(x) - 1$, the sign is practically always negative and the chance of hitting zero is infinitesimal.

A: You can construct an appropriate event function for extrema, where $y'(x) = 0$, by noting that $y'(x) = \cos(x)$. Then it is possible to use the **Direction** option to restrict the detection to points corresponding to a maximum.

```
NDSolve[{y'(x) == cos(x), y(0) == 0}, y, {x, 0, 20},
    Method → {EventLocator, "Event" → cos(x), "EventAction" :→ Sow[x],
        "Direction" → −1}] // Reap
```

$$\begin{pmatrix} \{y \to \text{InterpolatingFunction}[]\} \\ \{1.5708, 7.85398, 14.1372\} \end{pmatrix}$$

■ Operational Solutions to Differential Equations

Q: If $\phi(z)$ is the electrostatic potential on the axis of a cylindrically symmetric system, then the potential at the point (ρ, z), where ρ is the perpendicular distance from the axis, is given by the following (see, e.g., [6])

$$\phi(\rho, z) = J_0\left(\rho \frac{\partial}{\partial z}\right)\phi(z).$$

Q: How can I implement the operator $J_0\left(\rho \frac{\partial}{\partial z}\right)$?

A: The definition of $J_\nu(z)$ is given at functions.wolfram.com/03.01.02.0001.01.

$$J_\nu(z) = \sum_{k=0}^{\infty} \frac{(-1)^k \left(\frac{z}{2}\right)^{2k+\nu}}{\Gamma(k+\nu+1)\, k!}$$

True

A: Then, formally, one has

$$J_0\left(\rho \frac{\partial}{\partial z}\right) \equiv \sum_{k=0}^{\infty} \frac{(-1)^k \left(\rho \frac{\partial}{\partial z}\right)^{2k}}{4^k\, k!^2} = \sum_{k=0}^{\infty} \frac{1}{k!^2}\left(-\frac{\rho^2}{4} \frac{\partial^2}{\partial z^2}\right)^k.$$

A: For an arbitrary potential $\phi(z)$, the operator formalism can be used to obtain the Taylor series expansion of $\phi(\rho, z)$ about $\rho = 0$ using **NestList**. For example, here are the first four terms.

$$k = 0;\ \phi_4(\rho\_, z\_) = \text{Tr}@\text{NestList}\left[\left(++k;\ -\frac{\rho^2}{4k^2}\,\partial_{z,z}\#\right)\&,\ \phi(z),\ 3\right]$$

$$-\frac{\rho^6 \phi^{(6)}(z)}{2304} + \frac{1}{64}\rho^4 \phi^{(4)}(z) - \frac{1}{4}\rho^2 \phi''(z) + \phi(z)$$

A: For an axially symmetric potential, the Laplacian in cylindrical coordinates reads,

$$\nabla^2 \equiv \Delta = \frac{1}{\rho}\left(\frac{\partial}{\partial \rho}\left(\rho \frac{\partial}{\partial \rho}\right)\right) + \frac{\partial^2}{\partial z^2},$$

A: and the potential satisfies Laplace's equation, $\Delta V(\rho, z) = 0$. Verifying that the operator expansion produces a formal power-series solution is immediate.

$$\frac{\partial^2 \phi_4(\rho, z)}{\partial z^2} + O[\rho]^6$$

$$\phi''(z) - \frac{1}{4}\rho^2 \phi^{(4)}(z) + \frac{1}{64}\rho^4 \phi^{(6)}(z) + O(\rho^6)$$

$$\text{Collect}\left[\frac{1}{\rho}\left(\partial_\rho\left(\rho \frac{\partial \phi_4(\rho, z)}{\partial \rho}\right)\right), \rho, \text{Simplify}\right]$$

$$-\frac{1}{64}\rho^4 \phi^{(6)}(z) + \frac{1}{4}\rho^2 \phi^{(4)}(z) - \phi''(z)$$

% + %%

$$O(\rho^6)$$

A: Now for a concrete example. For a disk of radius R, with uniform surface charge density σ, oriented with its normal vector along the z-axis, here is the potential at $(0, 0, z)$.

$$V_{R\_}(z\_) = \text{Assuming}\left[z > R > 0, \frac{\sigma}{4\pi\epsilon_0}\int_0^{2\pi}\int_0^R \frac{1}{\sqrt{s^2+z^2}}\, s\, ds\, d\phi\right]$$

$$\frac{\sigma\left(\sqrt{R^2+z^2} - z\right)}{2\epsilon_0}$$

A: Using the operator formalism one obtains

$$k = 0;\ V_{3,R}(\rho\_, z\_) = \text{Simplify /@ Tr @ NestList}\left[\left(++k;\ -\frac{\rho^2}{4k^2}\partial_{z,z}\#\right)\&, V_R(z), 2\right]$$

$$-\frac{\rho^2 R^2 \sigma}{8\epsilon_0 (R^2+z^2)^{3/2}} + \frac{\sigma\left(\sqrt{R^2+z^2} - z\right)}{2\epsilon_0} - \frac{3\rho^4 \sigma (R^4 - 4R^2 z^2)}{128\epsilon_0 (R^2+z^2)^{7/2}}$$

A: There is another approach to this problem: In the case of azimuthal symmetry, the general solution to Laplace's equation $\Delta V = 0$ is (the multipole expansion),

$$V(r,\theta) = \sum_{l=0}^{\infty} \left(a_l\, r^l + \frac{b_l}{r^{l+1}} \right) P_l(\cos(\theta)), \tag{1}$$

A: where r and θ are the radial and polar spherical coordinates, respectively. Here is the truncated solution.

$$\mathbf{V_{n\_}(r\_,\,\theta\_) := \sum_{l=0}^{n} \left(a_l\, r^l + \frac{b_l}{r^{l+1}} \right) P_l(\cos(\theta))}$$

A: Now, if the potential is known on the axis, that is $V(r, 0) \equiv V(z)$, then one can use equation (0) to determine a_l and b_l by series expansion of $V(z)$ and term-by-term comparison. For $R > r > 0$, here is the series expansion of the axial potential.

$$\mathbf{Simplify}\left[V_R(r) + O[r]^5,\, R > 0\right]$$

$$\frac{R\,\sigma}{2\,\epsilon_0} - \frac{r\,\sigma}{2\,\epsilon_0} + \frac{r^2\,\sigma}{4\,R\,\epsilon_0} - \frac{r^4\,\sigma}{16\left(R^3\,\epsilon_0\right)} + O\left(r^5\right)$$

A: Now, equate this to $V(r, 0)$ and solve.

$$\mathbf{V_4(r, 0) + O[r]^5}$$

$$\frac{b_4}{r^5} + \frac{b_3}{r^4} + \frac{b_2}{r^3} + \frac{b_1}{r^2} + \frac{b_0}{r} + a_0 + a_1\,r + a_2\,r^2 + a_3\,r^3 + a_4\,r^4 + O\left(r^5\right)$$

Solve[% == %%, Join[Table[a_l, {l, 0, 4}], Table[b_l, {l, 0, 4}]]]

$$\left\{\left\{a_0 \to \frac{R\,\sigma}{2\,\epsilon_0},\, a_1 \to -\frac{\sigma}{2\,\epsilon_0},\, a_2 \to \frac{\sigma}{4\,R\,\epsilon_0},\, a_3 \to 0,\right.\right.$$
$$\left.\left. a_4 \to -\frac{\sigma}{16\,R^3\,\epsilon_0},\, b_0 \to 0,\, b_1 \to 0,\, b_2 \to 0,\, b_3 \to 0,\, b_4 \to 0\right\}\right\}$$

A: Hence we obtain the (truncated series expansion of the) potential of the disk off the axis in spherical coordinates.

$$\mathbf{V_4(r, \theta)\ /.\ First[\%]}$$

$$-\frac{r^4\,\sigma\left(35\cos^4(\theta) - 30\cos^2(\theta) + 3\right)}{128\,R^3\,\epsilon_0} + \frac{r^2\,\sigma\left(3\cos^2(\theta) - 1\right)}{8\,R\,\epsilon_0} - \frac{r\,\sigma\cos(\theta)}{2\,\epsilon_0} + \frac{R\,\sigma}{2\,\epsilon_0}$$

A: To compare this solution to that obtained earlier, we expand $\sqrt{R^2+r^2} = R\sqrt{1+(r/R)^2}$ into a series in r/R, valid for $0 < r < R$. See functions.wolfram.com/01.01.06.0002.01 and functions.wolfram.com/01.01.06.0003.01.

$$\sqrt{1+x} = {}_1F_0\left(-\frac{1}{2};;-x\right) = \sum_{n=0}^{\infty} \frac{(-1)^n \left(-\frac{1}{2}\right)_n}{n!} x^n$$

True

A: Check that we have the correct expansion for $V_R(r)$.

$$V_R(r) = \text{FullSimplify}\left[\frac{\sigma}{2\,\epsilon_0}\left(R\sum_{n=0}^{\infty}\frac{(-1)^n\left(-\frac{1}{2}\right)_n}{n!}\left(\frac{r}{R}\right)^{2n} - r\right), 0 < r < R\right]$$

True

A: Hence the potential of the disk off the axis is given by

$$\frac{\sigma}{2\,\epsilon_0}\left(R\sum_{n=0}^{\infty}\frac{(-1)^n\left(-\frac{1}{2}\right)_n}{n!}\left(\frac{r}{R}\right)^{2n} P_{2n}(\cos(\theta)) - r\cos(\theta)\right).$$

A: Changing coordinates from polar to cylindrical coordinates, $z = r\cos(\theta)$ and $r^2 = z^2 + \rho^2$, we verify that the two expansions are consistent.

$$\text{Simplify}\left[R\sum_{n=0}^{3}\frac{(-1)^n\left(-\frac{1}{2}\right)_n\left(\frac{r}{R}\right)^{2n} P_{2n}(\cos(\theta))}{n!} - r\cos(\theta)\ /.\cos(\theta)\to\frac{z}{r}\ /.\ r\to\sqrt{z^2+\rho^2}\right] + O[\rho]^5$$

$$\left(\frac{z^6}{16\,R^5} - \frac{z^4}{8\,R^3} + \frac{z^2}{2\,R} + R - z\right) + \frac{\rho^2\left(-8\,R^4 + 12\,R^2\,z^2 - 15\,z^4\right)}{32\,R^5} + \rho^4\left(\frac{45\,z^2}{128\,R^5} - \frac{3}{64\,R^3}\right) + O\left(\rho^5\right)$$

$$\text{Simplify}\left[\text{Series}\left[\% - \frac{2\,\epsilon_0}{\sigma}V_{3,R}(\rho, z), \{z, 0, 3\}\right], R > 0\right] \text{ // Normal}$$

0

A: Elmar Zeitler (zeitler@fhi-berlin.mpg.de) submitted another example of an operator expansion. Using the integral definition (functions.wolfram.com/03.01.07.0005.01),

$$J_n(z) = \frac{(-i)^n}{\pi} \int_0^{\pi} e^{i z \cos(\theta)} \cos(n\,\theta)\, d\theta,$$

A: and the identity (functions.wolfram.com/01.07.16.0096.01),

$$\cos(n\,\theta) = T_n(\cos(\theta)),$$

A: then the change of variables $\cos(\theta) \to x$ yields

$$J_n(z) = \frac{(-i)^n}{\pi} \int_{-1}^{1} \frac{e^{i z x}}{\sqrt{1-x^2}} T_n(x)\, dx.$$

A: Now, $T_n(x)$ is a polynomial in x and, since $x^k e^{i z x} = (-i)^k \partial_{\{z,k\}} e^{i z x}$, we see that

$$J_n(z) = \frac{(-i)^n}{\pi} T_n(-i\,\partial_z) \int_{-1}^{1} \frac{e^{i z x}}{\sqrt{1-x^2}}\, dx = (-i)^n\, T_n(-i\,\partial_z)\, J_0(z).$$

A: Implementation of this operator expansion is direct.

```
Table[{n, (-i)^n (Expand[z T_n(-i z)] /. z^k_. :> ∂_{z,k-1} #) &[J_0(z)] // Simplify},
  {n, 0, 4}]
```

$$\begin{pmatrix} 0 & J_0(z) \\ 1 & J_1(z) \\ 2 & J_2(z) \\ 3 & J_3(z) \\ 4 & J_4(z) \end{pmatrix}$$

■ Cluster Analysis

Q: For an arbitrary matrix of non-negative integers, how can I obtain the sum of those matrix elements that are surrounded by zeros? As a concrete example, for the following matrix an output of {4, 5, 7, 5, 4} is required.

$$\textbf{mat} = \begin{pmatrix} 0 & 0 & 0 & 1 & 0 & 0 & 0 & 5 & 0 & 0 \\ 0 & 0 & 0 & 3 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 1 & 3 & 0 & 0 & 0 \\ 1 & 2 & 2 & 0 & 0 & 0 & 1 & 0 & 0 & 3 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix};$$

Q: The order in which the groups surrounded by zero is summed does not matter.

A: Carl Woll (carlw@wolfram.com) answers: First, use **SparseArray** to get the positions of non-zero elements.

```
datarules = Most[ArrayRules[SparseArray[mat]]]
```

```
{{1, 4} → 1, {1, 8} → 5, {2, 4} → 3, {3, 2} → 1, {3, 7} → 1, {3, 8} → 3, {4, 1} → 1,
  {4, 2} → 2, {4, 3} → 2, {4, 7} → 1, {4, 10} → 3, {5, 3} → 1, {5, 10} → 1}
```

A: Next, define a distance function yielding 0 for identical elements, 1 for adjacent elements, and a big number, say 10, for nonadjacent elements.

```
myDistance[d1_, d2_] := Clip[Total[|d1 – d2|], {0, 1}, {10, 10}]
```

A: Then, use **FindClusters** with the **Agglomerate** method.

```
FindClusters[datarules, Method → Agglomerate, DistanceFunction → myDistance]
```

```
{{1, 3}, {5}, {1, 1, 2, 2, 1}, {1, 3, 1}, {3, 1}}
```

A: Finally, total the cluster values.

```
Total /@ %
```

```
{4, 5, 7, 5, 4}
```

A: Here is a function to do all the steps.

```
ClusterSums[array_] := Module[{datarules, clusteredvalues},
    datarules = Most[ArrayRules[SparseArray[array]]];
    clusteredvalues = FindClusters[datarules, Method → Agglomerate,
        DistanceFunction → myDistance];
    Total /@ clusteredvalues]
```

A: Check that **ClusterSums** works on **mat**.

```
ClusterSums[mat]
```

```
{4, 5, 7, 5, 4}
```

A: As a bonus, this approach can be extended to handle higher dimensional arrays.

■ References

[1] T. Wickham-Jones, *Mathematica Graphics: Techniques & Applications*, New York: TELOS/Springer-Verlag, 1994.

[2] library.wolfram.com/infocenter/Books/3753.

[3] E. Castillo, J. M. Gutiérrez, and A. Iglesias, "Solving a Functional Equation," *The Mathematica Journal*, **5**(1), 1995 pp. 82–86.

[4] library.wolfram.com/infocenter/Articles/857.

[5] library.wolfram.com/infocenter/Conferences/4760.

[6] E. Zeitler, "Operators in Electron Optics," *Optik* **3**, 1995 pp. 103–134.

P. Abbott, "In and Out," *The Mathematica Journal*, 2012. dx.doi.org/10.3888/tmj.10.2-1.

About the Author

Paul Abbott
School of Physics, M013
The University of Western Australia
35 Stirling Highway
Crawley WA 6009, Australia
tmj@physics.uwa.edu.au
physics.uwa.edu.au/~paul